
PROC PRESENT: Write Your Presentation in SAS® in as Few as Five Easy Steps
Louise S. Hadden, Abt Associates Inc., Cambridge, MA

ABSTRACT

There is no reason to wait for SAS® Version 9.3 and ODS PPT to create your presentations in SAS. Harness the
power of ODS PDF, PROC TEMPLATE and PROC REPORT (as well as any graphic producing procedure) in SAS
9.2 to create eye-popping and easy presentations, without ever leaving your SAS program! The presentation of the
paper will be written in SAS.

INTRODUCTION

SAS provides many tools for customizing SAS output. Although the use of SAS to create a presentation is
somewhat unusual, this methodology provides a brief look at some tools that can be used to enhance your
programming output, and demystify some procedures that might seem scary (i.e. PROC TEMPLATE!) All code was
run using PC SAS 9.2 Maintenance Release 3.

The idea for this project came about as a result of a “perfect SAS storm.” My sometime partner-in-crime, Mike Zdeb,
wrote a part of a joint presentation using PDF a few years back, which intrigued me. Around the same time, I began
using ODS PDF to create data-driven documents (example below.)

Then I became slightly obsessed with trying to make a Powerpoint presentation using ODS TAGSETS.RTF with
limited success… and then I saw Eric Gebhart’s Jabberwocky presentation on In-line Formatting. All these
elements came together, and using Tenniel illustrations and the text of Jabberwocky by Lewis Carroll a SAS PDF
presentation was born.

THE BACKGROUND GRAPHIC(S)

The choice of background graphics is an important one when created presentations in SAS as well as in PowerPoint
presentations. These backgrounds can be one and the same. Most presentations are in 8 ½ by 11 landscape
format, thus, it is important to use a background graphic as close to this size and orientation as possible. Many
software packages allow you to export JPEGs, and you can use existing JPEGs as backgrounds as well. For
example, if there is a Powerpoint presentation template, you can export the relevant page (empty) to a JPEG and
use that for your PDF presentation.

For this presentation, I created my background graphics using existing JPEGs and Microsoft Visio, exporting the
finished products to JPEGs.

The JPEG
shown on
the left is
one of the
samples
provided in
the zip file
in the
proceedings
and is the
background
for page 1
of the
example the
sample
program
creates. I
created
numbered
templates to
use with
identically
numbered
text fields.

THE PRESENTATION CONTENTS

Presentations generally consist of text, often bulleted or numbered, and graphics. Those in the “PowerPoint zone”
may add features such as links, notes, embedded sound, and animations. This paper will primarily address text and
graphics.

CREATING “PARAGRAPHS” IN SAS

SAS makes a very powerful and flexible desktop publisher. It is possible to construct very wide variables, so that
paragraphs can be created by concatenating text fields or variables using the CATT function. I use the CATT
function (which eliminates trailing blanks) so that I can string “sentences” together with a leading blank. I use the
escapechar n function to create line feeds. In the example below, the paragraphs are created in a data step, but you
could also import a file with existing fields. I label the variables using the “hidden dragon” (ALT+255), which
suppresses the printing of the label or variable name. The paragraphs can be printed using PROC REPORT with a
template without borders (and the special backgrounds we created above).

/* create paragraphs to print in presentation */

data test;

length text1 … text11 text12 $ 500;

blank = ' ';

text1a = '^n^nTwas brillig, and the slithy toves';

text1b = '^nDid gyre and gimble in the wabe:';

text1c = '^nAll mimsy were the borogoves,';

text1d = '^nAnd the mome raths outgrabe.';

text1=catt(text1a,text1b,text1c,text1d);

…

label blank = " " text1 = " " …;

run;

CREATING GRAPHICS IN YOUR PRESENTATION

There are many graphic producing procedures in SAS. SAS/GRAPH, the new SG Procedures, and statistical graphs
can all create images which can be directly inserted in your presentation, or imported into text fields and output using
ODS ESCAPECHAR, ODS TEXT, TITLE and FOOTNOTE statements, and some procedural statements.

MODIFYING A STYLE TEMPLATE

The choice of background graphics is an important one when created presentations in SAS as well as in PowerPoint
presentations. The example below demonstrates creating separate templates for each page in a macro, using
different graphics files as the background for each template. In order to get the correct “margins” for the template, it
was necessary to play around a little with the margins to get a full screen. As you can see, the “parent” for the
templates is the “noborder” template (code shown directly below), which does not include any lines in the output.
This template is useful for many purposes.

/* the noborder template */

proc template;

define style styles.noborder;

parent=styles.minimal;

style Table /

rules = NONE

frame = VOID

cellpadding = 0

cellspacing = 0

borderwidth = 0;

end;

run;

/* macro to create multiple style templates with different graphic backgrounds */

options orientation=landscape

nodate nonumber nocenter

leftmargin=.01in rightmargin=.01in

topmargin=.00001in bottommargin=.01in;

run;

ods listing close;

%macro maketem(num);

proc template;

define style styles.alice&num.;

parent=Styles.noborder;

class Body / backgroundimage="alice&num..jpg";

end;

run;

%mend;

PUTTING IT ALL TOGETHER

Using the style templates created above with the different backgrounds, I used PROC REPORT to print a blank
column (where the black background with the picture is) and a second column that prints in black type on the pink
background. The paragraph “wraps” in this second column. Note that you must use OPTIONS
ORIENTATION=LANDSCAPE to get the proper orientation on the page.

%macro rptit(num);

ods pdf style=styles.alice&num.;

proc report nowd data=test

style(report)=[cellpadding=3pt vjust=b]

style(header)=[just=center font_face='New Century Schoolbook'

font_weight=bold font_size=10pt]

style(lines)=[just=left font_face='New Century Schoolbook'] ;

columns blank text&num ;

define blank / style(COLUMN)={just=l font_face='New Century Schoolbook'

font_weight=bold

font_size=36pt cellwidth=600 }

;

define text&num / style(COLUMN)={just=l font_face='New Century Schoolbook'

font_weight=bold font_size=36pt cellwidth=800 }

;

run;

%mend;

ods pdf file='testalice.pdf' notoc ;

ods escapechar='^';

%rptit(1);

%rptit(2);

%rptit(3);

%rptit(4);

%rptit(5);

%rptit(6);

%rptit(7);

%rptit(8);

%rptit(9);

%rptit(10);

%rptit(11);

%rptit(12);

ods pdf close;

quit;

DISPLAYING THE PRESENTATION

A presentation created in SAS for the PDF destination can be viewed with Adobe Reader®, which is free and readily
available for many platforms. Once the presentation has been opened in Reader, choose View->Full Screen Mode
(CNTL-L) and move about the presentation as if you were in a Powerpoint presentation.

CONCLUSION

SAS provides us with a wide variety of tools and output destinations so that we can create exactly the output we
need for a given task. Using PROC TEMPLATE, PROC REPORT, and the ODS PDF destination presentations can
be written entirely in SAS. These same techniques can be used for a variety of purposes.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Louise Hadden louise_hadden@abtassoc.com

Code samples are available upon request, and a zip file with a program and graphics to create a PDF presentation
are included in the proceedings.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS
Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

